
APPENDIX 2:_MPLUS SCRIPT FOR ACE MODEL WITH GXE INVOLVING A FAMILY-LEVEL ENVIRONMENTAL VARIABLE (FIGURE 2)
title: 

ACE model with GxE involving a family-level environmental variable (see Figure 2)

In an ACE model with GxE, the phonotypic variance and covariance depend on the family-level moderator variable fv.

    The variance is 

    v = (a + ba*fv)**2 + (c + bc*fv)**2 + (e + be*fv)**2

    and the covariances are

    covmz = (a + ba*fv)**2 + (c + bc*fv)**2 for the MZ

    covdz = 0.5*(a + ba*fv)**2 + (c + bc*fv)**2 for the DZ

    where

    a = additive genetic effect

    ba = interaction of A with fv

    c = shared environment effect

    bc = interaction of C with fv

    e = non shared environment effect

    be = interaction of E with fv

    ** = exponentiation symbol.

    Note: Exclamation marks are for comments and ignored by Mplus.

data:

    file is IQData.dat;

variable:

    names are famnum zyg

              iq1 iq2 

              ses;

    missing are all (99);           !   Specify missing data symbol.

    usevariables are y1 y2 fv;

    constraint = fv;

    grouping is zyg (1=mz 2=dz);    !   Identify the grouping variable.

define:

    !

    !   Define the variables to be analyzed.

    !   y1 : phenotype measure of twin 1

    !   Y2 : phenotype measure of twin 2

    !

    y1=iq1;

    y2=iq2;

    !

    fv=ses;    !   family-level environmental variable

analysis: 

    type = general;

model:

    !

    !   The overall model.

    !   We estimate:

    !

    [y1]; [y2];     !   Means

    y1; y2;         !   Variances

    y1 with y2;     !   Covariance

    !

    y1 on fv(s);            !   Regression of phenotype 1 on the family variable

    y2 on fv(s);            !   Regression of phenotype 2 on the family variable

model mz:

    !

    !   The specific model for each group.

    !   Assign names to the parameters.

    !   The parameter names are provided in parentheses.

    !   Same name = equal parameters.

    !

    [y1](m); [y2](m);       !   Means are equal across twins and across MZ and DZ.  

    y1(v); y2(v);           !   Variances are equal across twins and across MZ and DZ.

    y1 with y2(covmz);      !   Covariance in the MZ group.

    y1 on fv(s);            !   Regression of phenotype 1 on the family variable

    y2 on fv(s);            !   Regression of phenotype 2 on the family variable

                            !   The coefficients are equal across twins and across 

                            !   MZ and DZ.

model dz:

    [y1](m); [y2](m);

    y1(v); y2(v);

    y1 with y2(covdz);      !   Covariance in the DZ group.

    y1 on fv(s);

    y2 on fv(s);

model constraint:

    !

    !   Define the model equations.

    !

    new(a ba c bc e be);      !   The names of the new parameters.

    v = (a + ba*fv)**2 + (c + bc*fv)**2 + (e + be*fv)**2;

    covmz = (a + ba*fv)**2 + (c + bc*fv)**2;

    covdz = 0.5*(a + ba*fv)**2 + (c + bc*fv)**2;

output:

    cinterval;      !   Request 95% and 99% confidence intervals.
